Identifying Gaps in Time Series Data
Bruce Gilsen, Federal Reserve Board, Washington, DC

ABSTRACT

Missing observations are a common issue with time series data. For a small amount of data, you can print the data set and inspect manually, but this is not realistic for large data sets, especially if there are multiple BY groups.

With the TIMEID procedure, which is part of SAS/ETS ® software, you can easily check a data set to determine if observations are missing, and print a simple report that shows the location and quantity of missing observations.

PROC TIMEID can be used with SAS ® date or datetime variables, and for any frequency. The examples in this paper use SAS dates with frequency WEEKDAY, which is common at the Federal Reserve Board.

Example 1. Data without a BY variable

Data set ONE contains WEEKDAY data, and DATE is a SAS date. Note that

- DATE, INCOME, INVEST, GOV, and TAXES are numeric variables. DATE contains SAS dates.
- April 17, 1995 is a Monday.
- 2 dates are missing after Tuesday, April 25.
- 1 date is missing after Friday, April 28.
- 4 dates are missing after Wednesday, May 3.

<table>
<thead>
<tr>
<th>Obs</th>
<th>date</th>
<th>income</th>
<th>invest</th>
<th>gov</th>
<th>taxes</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>19950417</td>
<td>1</td>
<td>2</td>
<td>3</td>
<td>4</td>
</tr>
<tr>
<td>2</td>
<td>19950418</td>
<td>5</td>
<td>6</td>
<td>7</td>
<td>8</td>
</tr>
<tr>
<td>3</td>
<td>19950419</td>
<td>9</td>
<td>10</td>
<td>11</td>
<td>12</td>
</tr>
<tr>
<td>4</td>
<td>19950420</td>
<td>13</td>
<td>14</td>
<td>15</td>
<td>16</td>
</tr>
<tr>
<td>5</td>
<td>19950421</td>
<td>17</td>
<td>18</td>
<td>19</td>
<td>20</td>
</tr>
<tr>
<td>6</td>
<td>19950424</td>
<td>21</td>
<td>22</td>
<td>23</td>
<td>24</td>
</tr>
<tr>
<td>7</td>
<td>19950425</td>
<td>25</td>
<td>26</td>
<td>27</td>
<td>28</td>
</tr>
<tr>
<td>8</td>
<td>19950428</td>
<td>29</td>
<td>30</td>
<td>31</td>
<td>32</td>
</tr>
<tr>
<td>9</td>
<td>19950502</td>
<td>33</td>
<td>34</td>
<td>35</td>
<td>36</td>
</tr>
<tr>
<td>10</td>
<td>19950503</td>
<td>37</td>
<td>38</td>
<td>39</td>
<td>40</td>
</tr>
<tr>
<td>11</td>
<td>19950510</td>
<td>41</td>
<td>42</td>
<td>43</td>
<td>44</td>
</tr>
</tbody>
</table>

Sort the data by the time ID variable if it is not already sorted that way. In this case, the sort shown here can be omitted.

```sas
proc sort data=one;
  by date;
run;
```

Now, PROC TIMEID examines data set ONE and creates data set OUTINT with information about missing observations. SAS dates are displayed with the YYMMDDN8. format.

```sas
proc timeid data=one outintervaldetails=outint;
  id date interval=weekday format=ymmddn8.;
run;
```
In SAS 9.3 and 9.4, the output data set has one observation and 38 variables. For readability, let's just print some useful variables.

```sas
proc print data=outint;
   var timeid start end nspans minspan maxspan msg interval;
run;
```

<table>
<thead>
<tr>
<th>Variable</th>
<th>Value</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>TIMEID</td>
<td>date</td>
<td>Name of the time ID variable</td>
</tr>
<tr>
<td>START</td>
<td>19950417</td>
<td>The smallest (earliest) time ID</td>
</tr>
<tr>
<td>END</td>
<td>19950510</td>
<td>The largest (latest) time ID</td>
</tr>
<tr>
<td>NSPANS</td>
<td>4</td>
<td>The number of spans in the data (1 = all values are contiguous, 2 = one gap in the data, 3 = 2 gaps in the data, etc.)</td>
</tr>
<tr>
<td>MINSPAN</td>
<td>1</td>
<td>Minimum number of periods between time ID values</td>
</tr>
<tr>
<td>MAXSPAN</td>
<td>5</td>
<td>Maximum number of periods between time ID values</td>
</tr>
<tr>
<td>MSG</td>
<td>(see below)</td>
<td>A status message that is also printed to the SAS log</td>
</tr>
<tr>
<td>INTERVAL</td>
<td>WEEKDAY</td>
<td>The time interval that was specified</td>
</tr>
</tbody>
</table>

**MSG value:** WARNING: Interval consistent but data contains gaps.

Now, use the ODS OUTPUT statement with PROC TIMEID to create data set DECOMP with information about missing observations. The ODS OUTPUT statement specifies which table (component of the results) is written to a SAS data set. The Printed Tabular Output section of the PROC TIMEID documentation has a list of ODS table names for PROC TIMEID.

The ODS OUTPUT statement is only in effect for the procedure that follows it. ODS OUTPUT statement syntax is

```sas
ods output table-name = data-set-name;
```

In this case, the ODS OUTPUT statement creates a data set that includes the following variables.

- **TIME** is a SAS date or datetime containing the values of the time ID variable. We'll rename it to DATE in the next step for readability since that was the name of our time ID variable.

- **SPAN** is the number of intervals between the current and previous values of TIME.
  - If one, observations are contiguous.
  - If two, then one period is missing between the previous and current observations.
  - If three, then two periods are missing between the previous and current observations.
  - etc.

- **COUNT** is the number of occurrences of the current value of TIME. If there are no repeat dates, it is 1 in all cases.

```sas
ods output Decomposition=decomp;
proc timeid data=one outintervaldetails=outint print=values;
   id date interval=weekday format=yymmddn8.;
run;
```

I'm more interested in the number of missing observations than the interval size. For example, the interval between observations 7 (19950425) and 8 (19950428) is 3 periods, but I want to know that there are 2 missing periods between these observations. So, I subtract 1 from SPAN. In observations that are contiguous to the previous observation, SPAN is now 0. SPAN is missing in observation 1 so the IF statement avoids a warning message.

```sas
data decomp;
   set decomp (rename=(time=Date));
   if _n_ ne 1 then span=span-1;
run;
```
Now, print observations that are not contiguous to the previous observation. By applying the WHERE clause to the data set rather than subsetting in a previous step, the output includes the observation numbers of the complete data set, which is useful for finding locations where data are missing.

```sas
proc print data=decomp;
   var date span count;
   where span>0;
run;
```

Output is as follows.

<table>
<thead>
<tr>
<th>Obs</th>
<th>Date</th>
<th>Span</th>
<th>Count</th>
</tr>
</thead>
<tbody>
<tr>
<td>8</td>
<td>19950428</td>
<td></td>
<td>1</td>
</tr>
<tr>
<td>9</td>
<td>19950502</td>
<td></td>
<td>1</td>
</tr>
<tr>
<td>11</td>
<td>19950510</td>
<td></td>
<td>1</td>
</tr>
</tbody>
</table>

We can easily see that observations are missing as follows.

- 2 dates are missing before April 28, 1995, which is observation 8.
- 1 date is missing before May 2, 1995, which is observation 9.
- 4 dates are missing before May 10, 1995, which is observation 11.

**Example 2. Data with a BY variable**

In many cases, time series is organized in BY groups. Let's create and then sort data set TWO, which includes two BY groups.

As before, DATE, INCOME, INVEST, GOV, and TAXES are numeric variables. DATE contains SAS dates and April 17, 1995 is a Monday.

- **For BANKID = 1:**
  - We have the same data values as in Example 1.
  - 2 dates are missing after Tuesday, April 25.
  - 1 date is missing after Friday, April 28.
  - 4 dates are missing after Wednesday, May 3.

- **For BANKID = 2:**
  - We have the same data values as in Example 1 except for an additional observation, for May 1.
  - 2 dates are missing after Tuesday, April 25.
  - 4 dates are missing after Wednesday, May 3.

<table>
<thead>
<tr>
<th>Obs</th>
<th>bankid</th>
<th>date</th>
<th>income</th>
<th>invest</th>
<th>gov</th>
<th>taxes</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>19950417</td>
<td>1</td>
<td>2</td>
<td>3</td>
<td>4</td>
</tr>
<tr>
<td>2</td>
<td>1</td>
<td>19950418</td>
<td>5</td>
<td>6</td>
<td>7</td>
<td>8</td>
</tr>
<tr>
<td>3</td>
<td>1</td>
<td>19950419</td>
<td>9</td>
<td>10</td>
<td>11</td>
<td>12</td>
</tr>
<tr>
<td>4</td>
<td>1</td>
<td>19950420</td>
<td>13</td>
<td>14</td>
<td>15</td>
<td>16</td>
</tr>
<tr>
<td>5</td>
<td>1</td>
<td>19950421</td>
<td>17</td>
<td>18</td>
<td>19</td>
<td>20</td>
</tr>
<tr>
<td>6</td>
<td>1</td>
<td>19950424</td>
<td>21</td>
<td>22</td>
<td>23</td>
<td>24</td>
</tr>
<tr>
<td>7</td>
<td>1</td>
<td>19950425</td>
<td>25</td>
<td>26</td>
<td>27</td>
<td>28</td>
</tr>
<tr>
<td>8</td>
<td>1</td>
<td>19950428</td>
<td>29</td>
<td>30</td>
<td>31</td>
<td>32</td>
</tr>
<tr>
<td>9</td>
<td>1</td>
<td>19950502</td>
<td>33</td>
<td>34</td>
<td>35</td>
<td>36</td>
</tr>
<tr>
<td>10</td>
<td>1</td>
<td>19950503</td>
<td>37</td>
<td>38</td>
<td>39</td>
<td>40</td>
</tr>
<tr>
<td>11</td>
<td>1</td>
<td>19950510</td>
<td>41</td>
<td>42</td>
<td>43</td>
<td>44</td>
</tr>
</tbody>
</table>
proc sort data=two;
  by bankid date;
run;

PROC TIMEID examines data set TWO and creates data set OUTINT with information about missing observations.

proc timeid data=two outintervaldetails=outint;
  id date interval=weekday format=yymmddn8.;
  by bankid;
run;
proc print data=outint;
  var bankid timeid start end nspans minspan maxspan msg interval;
run;

OUTINT contains 1 observation for each BY group.

<table>
<thead>
<tr>
<th>Obs</th>
<th>bankid</th>
<th>TIMEID</th>
<th>START</th>
<th>END</th>
<th>NSPANS</th>
<th>MINSPAN</th>
<th>MAXSPAN</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>date</td>
<td>19950417</td>
<td>19950510</td>
<td>4</td>
<td>1</td>
<td>5</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>date</td>
<td>19950417</td>
<td>19950510</td>
<td>3</td>
<td>1</td>
<td>5</td>
</tr>
</tbody>
</table>

Obs                  MSG                          INTERVAL
1    WARNING: Interval consistent but data contains gaps. WEEKDAY
2    WARNING: Interval consistent but data contains gaps. WEEKDAY

As in Example 1, PROC TIMEID with the ODS OUTPUT statement creates data set DECOMP with information about missing observations, and after subtracting 1 from SPAN it contains the number of missing periods between the current and previous observations. The IF statement avoids warning messages in the first observation of each BY group, when SPAN is missing.

ods output Decomposition=decomp;
proc timeid data=two outintervaldetails=outint print=values;
  id date interval=weekday format=yymmddn8.;
  by bankid;
run;

data decomp;
  set decomp (rename=(time=Date));
  by bankid;
  if not first.bankid then span=span-1;
run;
proc print data=decomp;
```sas
var bankid date span count;
where span>0;
run;
```

Output is as follows.

<table>
<thead>
<tr>
<th>Obs</th>
<th>bankid</th>
<th>Date</th>
<th>Span</th>
<th>Count</th>
</tr>
</thead>
<tbody>
<tr>
<td>8</td>
<td>1</td>
<td>19950428</td>
<td>2</td>
<td>1</td>
</tr>
<tr>
<td>9</td>
<td>1</td>
<td>19950502</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>11</td>
<td>1</td>
<td>19950510</td>
<td>4</td>
<td>1</td>
</tr>
<tr>
<td>19</td>
<td>2</td>
<td>19950428</td>
<td>2</td>
<td>1</td>
</tr>
<tr>
<td>23</td>
<td>2</td>
<td>19950510</td>
<td>4</td>
<td>1</td>
</tr>
</tbody>
</table>

We can easily see that observations are missing as follows.

- 2 dates are missing before April 28, 1995, which is observation 8.
- 1 date is missing before May 2, 1995, which is observation 9.
- 4 dates are missing before May 10, 1995, which is observation 11.
- 2 dates are missing before April 28, 1995 for BANKID 2, which is observation 19.
- 4 dates are missing before May 10, 1995 for BANKID 2, which is observation 23.

### A macro to generalize the code

Macro TSGAPS generalizes the code in the previous examples. It contains the code used in the previous two examples, with a few small additions.

- Conditional macro code generates slightly different SAS code for cases with and without BY variables.
- The final data set is checked to see if any observations are missing. If so, the report detailing missing observations is printed. Otherwise, a note is written to the SAS log and no report is printed.

Macro TSGAPS has four required arguments and two optional arguments, as follows.

```sas
%TSGAPS(data=, out=, timeid=, interval=<, byvar=>, format=>);
```

#### Required arguments

- **data=** *input-dataset*
  The data set to examine

- **out=** *output-dataset*
  The output data set containing the final results

- **timeid=** *timeid-variable*
  The time ID variable: a SAS date or datetime

- **interval=** *interval-value*
  The periodicity of the time ID variable

#### Optional arguments

- **byvar=** *byvar-values*
  A list of BY variables if any
Identifying Gaps in Time Series Data

**format=format-value**

Format used to display the time ID variable, defaults to YYMMDDN8.

Here is the macro.

```sas
%macro tsgaps(data=, out=, timeid=, interval=, byvar=, format=yymmddn8.);

   %local max_span;

   /* Verify that required arguments were provided */
   %if &data = or &out = or &timeid = or &interval = %then %do;
     %put ERROR: Macro TSGAPS aborts because required argument not provided.;
     %end;
   %else %do;

      /* Sort by the BY variables if any, and by the Time ID variable */
      proc sort data=&data;
         by &byvar &timeid;
      run;

      /* Examine the data set for missing observations and print some results */
      proc timeid data=&data outintervaldetails=outint;
         id &timeid interval=&interval format=&format;
         %if &byvar ne %then %do;
            by &byvar;
         %end;
      run;

      proc print data=outint;
         var &byvar timeid start end nspans minspan maxspan msg interval;
      run;

      /* Create a SAS data set with information about missing observations */
      ods output Decomposition=&out;
      proc timeid data=&data outintervaldetails=all print=values;
         id &timeid interval=&interval format=&format;
         %if &byvar ne %then %do;
            by &byvar;
         %end;
      run;

      /* Modify the data so that SPAN has the number of missing observations
       between the current and previous observation.  Print a summary
       report that shows the missing observations. */
      data &out;
      set &out (rename=(time=&timeid));
      %if &byvar ne %then %do;
         by &byvar;
         if not first.&byvar then span=span-1;
      %end;
      %else %do;
         if _n_ ne 1 then span=span-1;
      %end;
      run;
```
Identifying Gaps in Time Series Data

/* Determine largest value of SPAN and copy it to macro variable MAX_SPAN */
proc means data=&out noprint;
  var span;
  output out=_&out max=maxspan;
run;
proc sql noprint;
  select maxspan into :max_span trimmed
  from _&out;
quit;

/* If maximum value of SPAN is 0, then no observations are missing.
Otherwise, print a missing observation summary report. */
%if &max_span = 0 %then %do;
  %put NOTE: No observations in data set &data are missing;
%end;
%else %do;
  proc print data=&out;
    var &byvar &timeid span count;
    where span>0;
  run;
%end;
%mend tsgaps;

Here is how to invoke the macro for the examples in this paper.

For example 1:

    %tsgaps(data=one, out=decomp1, timeid=date, interval=weekday);

For example 2:

    %tsgaps(data=two, out=decomp2, timeid=date, interval=weekday, byvar=bankid);

Beyond the basics

The code in this paper shows a small subset of the capabilities of PROC TIMEID. It also reflects the personal preferences of the author, for example, to see the number of missing periods rather than the number of periods between two observations. I encourage you to read the PROC TIMEID documentation and design code that provides the information most useful to you.

CONCLUSION

PROC TIMEID, which is part of SAS/ETS software, can be used to obtain information about missing observations in time series data, which is a common problem. The paper contained a few simple examples, and the PROC TIMEID documentation includes information about other features that might be useful.

REFERENCES


ACKNOWLEDGMENTS

This paper was heavily influenced by Jennifer Beeman's SAS Communities article on PROC TIMEID (Beeman, 2014). The code used in Example 1 is taken from code in that article.

In addition, the following people contributed extensively to the development of this paper: Heidi Markovitz and Donna Hill at the Federal Reserve Board.

CONTACT INFORMATION

For more information, contact the author, Bruce Gilsen, by mail at Federal Reserve Board, Mail Stop N-122, Washington, DC 20551; by e-mail at bruce.gilsen@frb.gov; or by phone at 202-452-2494.

TRADEMARK INFORMATION

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.
Other brand and product names are registered trademarks or trademarks of their respective companies.